EXAMPLE 3.1-1: Design of a Conveyor Belt

Plastic parts are formed in an injection mold and dropped (flat) onto a conveyor belt (Figure 1). The parts are disk-shaped with thickness th = 2.0 mm and diameter D = 10.0 cm. The plastic has thermal conductivity k = 0.35 W/m-K, density ρ = 1100 kg/m^3, and specific heat capacity c = 1900 J/kg-K. The side of the part that faces the conveyor belt is adiabatic. The top surface of the part is exposed to air at T_∞ = 20°C with a heat transfer coefficient $\overline{h}$ = 15 W/m^2-K. The temperature of the part immediately after it is formed is T_{ini} = 180°C. The part must be cooled to T_{max} = 80°C before it can be stacked and packaged. The packaging system is positioned L = 15 ft away from the molding machine.



Figure 1: Plastic piece

a.) Is a lumped capacitance model of the part justified for this situation?

The inputs are entered in EES:

```
"EXAMPLE 3.1-1: Design of a Conveyor Belt"
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
th=2.0 [mm]*convert(mm,m)                    "thickness"
k=0.35 [W/m-K]                               "conductivity"
D=10 [cm]*convert(cm,m)                      "diameter"
rho=1100 [kg/m^3]                            "density"
c=1900 [J/kg-K]                              "specific heat capacity"
h_bar=15 [W/m^2-K]                           "heat transfer coefficient"
T_ini=converttemp(C,K,180 [C])               "initial temperature of part"
T_infinity=converttemp(C,K,20 [C])           "ambient temperature"
T_max=converttemp(C,K,80 [C])                "maximum handling temperature"
L=15 [ft]*convert(ft,m)                      "conveyor length"
```

The lumped capacitance model can be justified by examing the Biot number, the ratio of the resistance to internal conduction heat transfer to the resistance to heat transfer from the surface of the object. In this problem, the resistance to heat transfer from the surface is due only to convection and therefore Eq. (3-1) is valid, where the conduction length is intuitively the thickness of the object (it would be the half-width if the conveyor side of the part were not adiabatic). Note that the characteristic length defined by Eq. (3-2) as the ratio of the volume (V) to the exposed area for heat transfer (A_s) is also equal to the thickness of the part:

$$V = \frac{\pi D^2 th}{4}$$

$$A_s = \frac{\pi D^2}{4}$$

$$L_{cond} = \frac{V}{A_s} = \frac{\pi D^2 th}{4} \frac{4}{\pi D^2} = th$$

```
V=pi*D^2*th/4                    "Volume"
As=pi*D^2/4                      "Surface area exposed to cooling"
L_cond=V/As                      "Conduction length"
Bi=h_bar*L_cond/k                "Biot number based on conduction length"
```

The Biot number predicted by EES is 0.09 which is sufficiently small to use the lumped capacitance model unless very high accuracy is required.

b.) What is the maximum acceptable conveyor velocity so that the parts arrive at the packaging station below T_{max}?

The governing differential equation is obtained by considering a control volume that encloses the entire plastic part (see Figure 1); the energy balance suggested by the control volume is:

$$0 = \dot{q}_{conv} + \frac{dU}{dt}$$

The convection heat transfer is:

$$\dot{q}_{conv} = \bar{h} A_s \left(T - T_\infty \right)$$

and the rate of energy storage is:

$$\frac{dU}{dt} = \rho V c \frac{dT}{dt}$$

Combining these equations leads to:

$$0 = \bar{h} A_s \left(T - T_\infty \right) + \rho V c \frac{dT}{dt}$$

which can be rearranged:

$$\frac{dT}{dt} = -\frac{\left(T - T_{\infty}\right)}{\tau_{lumped}} \tag{1}$$

where τ_{lumped} is the time constant for this problem:

$$\tau_{lumped} = \frac{\rho V c}{\bar{h} A_s}$$

```
tau_lumped=V*rho*c/(h_bar*As)          "time constant"
```

The time constant for part is 279 s. We should keep in mind that it will take on the order of 5 minutes to cool the plastic piece substantially and use this insight to check any more precise solution that is obtained.

Equation (1) is a 1st order differential equation with the boundary condition:

$$T_{t=0} = T_{ini} \tag{2}$$

The differential equation is separable; that is, all of the terms involving the dependent variable, T, can be placed on one side while the terms involving the independent variable, t, can be placed on the other:

$$\frac{dT}{\left(T - T_{\infty}\right)} = -\frac{dt}{\tau_{lumped}}$$

The separated equation can be directly integrated:

$$\int_{T_{ini}}^{T} \frac{dT}{\left(T - T_{\infty}\right)} = -\int_{0}^{t} \frac{dt}{\tau} \tag{3}$$

This integration is most easily accomplished by defining the temperature difference (θ):

$$\theta = T - T_{\infty} \tag{4}$$

so that:

$$d\theta = dT \tag{5}$$

Substituting Eqs. (4) and (5) into Eq. (3) leads to:

$$\int_{T_{ini}-T_\infty}^{T-T_\infty} \frac{d\theta}{\theta} = -\int_0^t \frac{dt}{\tau_{lumped}}$$

Carrying out the integration leads to:

$$\ln\left(\frac{T-T_\infty}{T_{ini}-T_\infty}\right) = -\frac{t}{\tau_{lumped}}$$

Solving for T leads to:

$$T = T_\infty + \left(T_{ini}-T_\infty\right)\exp\left(-\frac{t}{\tau_{lumped}}\right) \quad (6)$$

which is equivalent to the entry in Table 3-1 for a step change in ambient temperature. The time required to cool the part from T_{ini} to T_{max} can be computed using Eq. (6):

```
T_max=T_infinity+(T_ini-T_infinity)*exp(-t_cool/tau_lumped)        "time required to cool part"
```

and is found to be t_{cool} = 273 s; note that this value is in good agreement with the previously calculated time constant. The linear velocity of the conveyor required so that it takes at least 273 s to travel the 15 ft between the molding machine and the packaging station is:

$$u_c = \frac{L}{t_{cool}}$$

```
u_c=L/t_cool                                   "conveyor velocity"
u_c_fpm=u_c*convert(m/s,ft/min)                "conveyor velocity in ft/min"
```

The maximum allowable velocity u_c is found to be 0.0167 m/s (3.29 ft/min).